NAG Fortran Library Routine Document

F08XPF (ZGGESX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08XPF (ZGGESX) computes the generalized eigenvalues, the generalized Schur form (S, T), and, optionally the left and/or right generalized Schur vectors for a pair of n by n complex nonsymmetric matrices (A, B).

Estimates of condition numbers for selected generalized eigenvalue clusters and Schur vectors are also computed.

2 Specification

```
      SUBROUTINE F08XPF (JOBVSL, JOBVSR, SORT, SELCTG, SENSE, N, A, LDA, B,
     1                   LDB, SDIM, ALPHA, BETA, VSL, LDVSL, VSR, LDVSR,
     2                   RCONDE, RCONDV, WORK, LWORK, RWORK, IWORK, LIWORK,
     3                   BWORK, INFO)
      INTEGER            N, LDA, LDB, SDIM, LDVSL, LDVSR, LWORK, IWORK(*),
     1                   LIWORK, INFO
      double precision   RCONDE(2), RCONDV(2), RWORK(*)
      complex*16         A(LDA,*), B(LDB,*), ALPHA(*), BETA(*), VSL(LDVSL,*),
     1                   VSR(LDVSR,*), WORK(*)
      LOGICAL            SELCTG, BWORK(*)
      CHARACTER*1        JOBVSL, JOBVSR, SORT, SENSE
      EXTERNAL           SELCTG
```

The routine may be called by its LAPACK name ***zggesx***.

3 Description

The generalized Schur factorization for a pair of complex matrices (A, B) is given by

$$A = QSZ^H, \quad B = QTZ^H,$$

where Q and Z are unitary, T and S are upper triangular. The generalized eigenvalues, λ, of (A, B) are computed from the diagonals of T and S and satisfy

$$Az = \lambda Bz,$$

where z is the corresponding generalized eigenvector. λ is actually returned as the pair (α, β) such that

$$\lambda = \alpha/\beta$$

since β, or even both α and β can be zero. The columns of Q and Z are the left and right generalized Schur vectors of (A, B).

Optionally, F08XNF (ZGGES) can order the generalized eigenvalues on the diagonals of (S, T) so that selected eigenvalues are at the top left. The leading columns of Q and Z then form an orthonormal basis for the corresponding eigenspaces, the deflating subspaces.

F08XNF (ZGGES) computes T to have real non-negative diagonal entries. The generalized Schur factorization, before reordering, is computed by the QZ algorithm.

The reciprocals of the condition estimates, the reciprocal values of the left and right projection norms, are returned in $\mathrm{RCONDE}(1)$ and $\mathrm{RCONDE}(2)$ respectively, for the selected generalized eigenvalues, together with reciprocal condition estimates for the corresponding left and right deflating subspaces, in $\mathrm{RCONDV}(1)$ and $\mathrm{RCONDV}(2)$. See Section 4.11 of Anderson *et al.* (1999) for further information.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBVSL – CHARACTER*1 *Input*

On entry: if JOBVSL = 'N', do not compute the left Schur vectors.

If JOBVSL = 'V', compute the left Schur vectors.

2: JOBVSR – CHARACTER*1 *Input*

On entry: if JOBVSR = 'N', do not compute the right Schur vectors.

If JOBVSR = 'V', compute the right Schur vectors.

3: SORT – CHARACTER*1 *Input*

On entry: specifies whether or not to order the eigenvalues on the diagonal of the generalized Schur form:

if SORT = 'N', eigenvalues are not ordered;
if SORT = 'S', eigenvalues are ordered (see SELCTG).

4: SELCTG – LOGICAL FUNCTION, supplied by the user. *External Procedure*

If SORT = 'S', SELCTG is used to select generalized eigenvalues to the top left of the generalized Schur form.

If SORT = 'N', SELCTG is not referenced and F08XPF (ZGGESX) may be called with the dummy function F08XNZ.

Its specification is:

```
LOGICAL FUNCTION SELCTG (A, B)
complex*16                A, B
```

1: A – ***complex*16*** *Input*
2: B – ***complex*16*** *Input*

On entry: an eigenvalue $\mathrm{A}(j)/\mathrm{B}(j)$ is selected if $\mathrm{SELCTG}(\mathrm{A}(j),\mathrm{B}(j))$ is true.

Note that in the ill-conditioned case, a selected generalized eigenvalue may no longer satisfy $\mathrm{SELCTG}(\mathrm{A}(j),\mathrm{B}(j))$ = .TRUE. after ordering. INFO is set to $\mathrm{N}+2$ in this case. (See INFO below).

SELCTG must be declared as EXTERNAL in the (sub)program from which F08XPF (ZGGESX) is called. Parameters denoted as *Input* must **not** be changed by this procedure.

5: SENSE – CHARACTER*1 *Input*

On entry: determines which reciprocal condition numbers are computed:

if SENSE = 'N', none are computed;
if SENSE = 'E', computed for average of selected eigenvalues only;
if SENSE = 'V', computed for selected deflating subspaces only;
if SENSE = 'B', computed for both.

If SENSE = 'E', 'V' or 'B', SORT must equal 'S'.

6: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

7: A(LDA,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the first of the pair of matrices, A.

On exit: has been overwritten by its generalized Schur form S.

8: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08XPF (ZGGESX) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

9: B(LDB,$*$) – ***complex*16*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{N})$.

On entry: the second of the pair of matrices, B.

On exit: has been overwritten by its generalized Schur form T.

10: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08XPF (ZGGESX) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \mathrm{N})$.

11: SDIM – INTEGER *Output*

On exit: if SORT $=$ 'N', $\mathrm{SDIM} = 0$.

If SORT $=$ 'S', SDIM $=$ number of eigenvalues (after sorting) for which SELCTG is true.

12: ALPHA($*$) – ***complex*16*** array *Output*

Note: the dimension of the array ALPHA must be at least $\max(1, \mathrm{N})$.

On exit: see the description of BETA below.

13: BETA($*$) – ***complex*16*** array *Output*

Note: the dimension of the array BETA must be at least $\max(1, \mathrm{N})$.

On exit: $\mathrm{ALPHA}(j)/\mathrm{BETA}(j)$, $j = 1, \ldots, \mathrm{N}$, will be the generalized eigenvalues. $\mathrm{ALPHA}(j)$ and $\mathrm{BETA}(j), j = 1, \ldots, \mathrm{N}$ are the diagonals of the complex Schur form (S, T). $\mathrm{BETA}(j)$ will be non-negative real.

Note: the quotients $\mathrm{ALPHA}(j)/\mathrm{BETA}(j)$ may easily over- or underflow, and $\mathrm{BETA}(j)$ may even be zero. Thus, the user should avoid naively computing the ratio α/β. However, ALPHA will be always less than and usually comparable with $\|\mathrm{A}\|$ in magnitude, and BETA always less than and usually comparable with $\|\mathrm{B}\|$.

14: VSL(LDVSL,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array VSL must be at least $\max(1, \mathrm{N})$.

On exit: if JOBVSL $=$ 'V', VSL will contain the left Schur vectors, Q.

If JOBVSL $=$ 'N', VSL is not referenced.

15: LDVSL – INTEGER *Input*

On entry: the first dimension of the array VSL as declared in the (sub)program from which F08XPF (ZGGESX) is called.

Constraints:

if JOBVSL = 'V', LDVSL $\geq \max(1, \mathrm{N})$;
LDVSL ≥ 1 otherwise.

16: VSR(LDVSR,$*$) – ***complex*16*** array *Output*

Note: the second dimension of the array VSR must be at least $\max(1, \mathrm{N})$.

On exit: if JOBVSR = 'V', VSR will contain the right Schur vectors, Z.

If JOBVSR = 'N', VSR is not referenced.

17: LDVSR – INTEGER *Input*

On entry: the first dimension of the array VSR as declared in the (sub)program from which F08XPF (ZGGESX) is called.

Constraints:

if JOBVSR = 'V', LDVSR $\geq \max(1, \mathrm{N})$;
LDVSR ≥ 1 otherwise.

18: RCONDE(2) – ***double precision*** array *Output*

On exit: if SENSE = 'E' or 'B', RCONDE(1) and RCONDE(2) contain the reciprocal condition numbers for the average of the selected eigenvalues.

If SENSE = 'N' or 'V', RCONDE is not referenced.

19: RCONDV(2) – ***double precision*** array *Output*

On exit: if SENSE = 'V' or 'B', RCONDV(1) and RCONDV(2) contain the reciprocal condition numbers for the selected deflating subspaces.

If SENSE = 'N' or 'E', RCONDV is not referenced.

20: WORK$(*)$ – ***complex*16*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) returns the optimal LWORK.

21: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08XPF (ZGGESX) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the bound on the optimal size of the WORK array and the minimum size of the IWORK array, returns these values as the first entries of the WORK and IWORK arrays and no error message related to LWORK or LIWORK is issued.

Constraints:

if N $= 0$, LWORK ≥ 1;
if SENSE = 'E', 'V' or 'B', LWORK $\geq \max(1, 2 \times \mathrm{N}, 2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM}))$;
LWORK $\geq \max(1, 2 \times \mathrm{N})$ otherwise.

Note: $2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM}) \leq \mathrm{N} \times \mathrm{N}/2$. Note also that an error is only returned if LWORK $< \max(1, 2 \times \mathrm{N})$, but if SENSE = 'E', 'V' or 'B' this may not be large enough. Consider increasing LWORK by nb, where nb is the block size.

22: RWORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array RWORK must be at least $\max(1, 8 \times \mathrm{N})$.

Real workspace.

23: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{LIWORK})$.

On exit: if INFO $= 0$, IWORK(1) returns the minimum LIWORK.

24: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08XPF (ZGGESX) is called.

If LIWORK $= -1$, a workspace query is assumed; the routine only calculates the bound on the optimal size of the WORK array and the minimum size of the IWORK array, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if SENSE $=$ 'N' or N $= 0$, LIWORK ≥ 1;
LIWORK $\geq \mathrm{N} + 2$ otherwise.

25: BWORK($*$) – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least 1 if SORT $=$ 'N' and at least $\max(1, \mathrm{N})$ otherwise.

If SORT $=$ 'N', BWORK is not referenced.

26: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO $= 1$ to N

The QZ iteration failed. (A, B) are not in Schur form, but ALPHA(j) and BETA(j) should be correct for $j = \mathrm{INFO} + 1, \ldots, \mathrm{N}$.

INFO $> \mathrm{N}$

$= \mathrm{N} + 1$: other than QZ iteration failed in F08XSF (ZHGEQZ)

$= \mathrm{N} + 2$: after reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues in the generalized Schur form no longer satisfy SELCTG = .TRUE.. This could also be caused due to scaling.

$= \mathrm{N} + 3$: reordering failed because some eigenvalues were too close to separate (the problem is very ill-conditioned).

7 Accuracy

The computed generalized Schur factorization satisfies

$$A + E = QSZ^T, \quad B + F = QTZ^T,$$

where

$$\|(E,F)\|_F = O(\epsilon)\|(A,B)\|_F$$

and ϵ is the ***machine precision***. See Section 4.11 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3.

The real analogue of this routine is F08XBF (DGGESX).

9 Example

To find the generalized Schur factorization of the matrix pair (A, B), where

$$A = \begin{pmatrix} -21.10 - 22.50i & 53.50 - 50.50i & -34.50 + 127.50i & 7.50 + 0.50i \\ -0.46 - 7.78i & -3.50 - 37.50i & -15.50 + 58.50i & -10.50 - 1.50i \\ 4.30 - 5.50i & 39.70 - 17.10i & -68.50 + 12.50i & -7.50 - 3.50i \\ 5.50 + 4.40i & 14.40 + 43.30i & -32.50 - 46.00i & -19.00 - 32.50i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 1.00 - 5.00i & 1.60 + 1.20i & -3.00 + 0.00i & 0.00 - 1.00i \\ 0.80 - 0.60i & 3.00 - 5.00i & -4.00 + 3.00i & -2.40 - 3.20i \\ 1.00 + 0.00i & 2.40 + 1.80i & -4.00 - 5.00i & 0.00 - 3.00i \\ 0.00 + 1.00i & -1.80 + 2.40i & 0.00 - 4.00i & 4.00 - 5.00i \end{pmatrix},$$

such that the eigenvalues of (A, B) for which $|\lambda| < 6$ correspond to the top left diagonal elements of the generalized Schur form, (S, T). Estimates of the condition numbers for the selected eigenvalue cluster and corresponding deflating subspaces are also returned.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08XPF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDB, LDVSL, LDVSR, LIWORK, LWORK
      PARAMETER        (LDA=NMAX,LDB=NMAX,LDVSL=NMAX,LDVSR=NMAX,
     +                 LIWORK=NMAX+2,LWORK=NMAX*NB+NMAX*NMAX/2)
*     .. Local Scalars ..
      DOUBLE PRECISION ABNORM, ANORM, BNORM, EPS, TOL
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N, SDIM
*     .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), ALPHA(NMAX), B(LDB,NMAX),
     +                 BETA(NMAX), VSL(LDVSL,NMAX), VSR(LDVSR,NMAX),
     +                 WORK(LWORK)
      DOUBLE PRECISION RCONDE(2), RCONDV(2), RWORK(8*NMAX)
      INTEGER          IWORK(LIWORK)
      LOGICAL          BWORK(NMAX)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Functions ..
      DOUBLE PRECISION F06BNF, F06UAF, X02AJF
      LOGICAL          SELCTG
      EXTERNAL         F06BNF, F06UAF, X02AJF, SELCTG
*     .. External Subroutines ..
```

```
      EXTERNAL         X04DBF, ZGGESX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08XPF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read in the matrices A and B
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         READ (NIN,*) ((B(I,J),J=1,N),I=1,N)
*
*        Find the Frobenius norms of A and B
*
         ANORM = F06UAF('Frobenius',N,N,A,LDA,RWORK)
         BNORM = F06UAF('Frobenius',N,N,B,LDB,RWORK)
*
*        Find the generalized Schur form
*
         CALL ZGGESX('Vectors (left)','Vectors (right)','Sort',SELCTG,
     +               'Both reciprocal condition numbers',N,A,LDA,B,LDB,
     +               SDIM,ALPHA,BETA,VSL,LDVSL,VSR,LDVSR,RCONDE,RCONDV,
     +               WORK,LWORK,RWORK,IWORK,LIWORK,BWORK,INFO)
*
         IF (INFO.GT.0 .AND. INFO.NE.(N+2)) THEN
            WRITE (NOUT,99999) 'Failure in ZGGESX. INFO =', INFO
         ELSE
            WRITE (NOUT,99999)
     +        'Number of eigenvalues for which SELCTG is true = ', SDIM,
     +        '(dimension of deflating subspaces)'
            WRITE (NOUT,*)
            IF (INFO.EQ.(N+2)) THEN
               WRITE (NOUT,99998) '***Note that rounding errors mean ',
     +           'that leading eigenvalues in the generalized',
     +           'Schur form no longer satisfy SELCTG = .TRUE.'
               WRITE (NOUT,*)
            END IF
*
*           Print out the factors of the generalized Schur factorization
*
            IFAIL = 0
            CALL X04DBF('General',' ',N,N,A,LDA,'Bracketed','F7.2',
     +                  'Generalized Schur matrix S','Integer',RLABS,
     +                  'Integer',CLABS,80,0,IFAIL)
*
            WRITE (NOUT,*)
            CALL X04DBF('General',' ',N,N,B,LDB,'Bracketed','F7.2',
     +                  'Generalized Schur matrix T','Integer',RLABS,
     +                  'Integer',CLABS,80,0,IFAIL)
*
            WRITE (NOUT,*)
            CALL X04DBF('General',' ',N,N,VSL,LDVSL,'Bracketed','F7.4',
     +                  'Matrix of left generalized Schur vectors',
     +                  'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
*
            WRITE (NOUT,*)
            CALL X04DBF('General',' ',N,N,VSR,LDVSR,'Bracketed','F7.4',
     +                  'Matrix of right generalized Schur vectors',
     +                  'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
*
*           Print out the reciprocal condition numbers
*
            WRITE (NOUT,*)
            WRITE (NOUT,99997)
     +        'Reciprocals of left and right projection norms onto',
     +        'the deflating subspaces for the selected eigenvalues',
     +        'RCONDE(1) = ', RCONDE(1), ', RCONDE(2) = ', RCONDE(2)
            WRITE (NOUT,*)
            WRITE (NOUT,99997)
```

```
     +        'Reciprocal condition numbers for the left and right',
     +        'deflating subspaces', 'RCONDV(1) = ', RCONDV(1),
     +        ', RCONDV(2) = ', RCONDV(2)
*
*           Compute the machine precision and sqrt(ANORM**2+BNORM**2)
*
            EPS = X02AJF()
            ABNORM = F06BNF(ANORM,BNORM)
            TOL = EPS*ABNORM
*
*           Print out the approximate asymptotic error bound on the
*           average absolute error of the selected eigenvalues given by
*
*              eps*norm((A, B))/PL,   where PL = RCONDE(1)
*
            WRITE (NOUT,*)
            WRITE (NOUT,99996)
     +        'Approximate asymptotic error bound for selected ',
     +        'eigenvalues    = ', TOL/RCONDE(1)
*
*           Print out an approximate asymptotic bound on the maximum
*           angular error in the computed deflating subspaces given by
*
*              eps*norm((A, B))/DIF(2),   where DIF(2) = RCONDV(2)
*
            WRITE (NOUT,99996)
     +        'Approximate asymptotic error bound for the deflating ',
     +        'subspaces = ', TOL/RCONDV(2)
*
            LWKOPT = WORK(1)
            IF (LWORK.LT.LWKOPT) THEN
               WRITE (NOUT,*)
               WRITE (NOUT,99995) 'Optimum workspace required = ',
     +           LWKOPT, 'Workspace provided         = ', LWORK
            END IF
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I4,/1X,A)
99998 FORMAT (1X,2A,/1X,A)
99997 FORMAT (1X,A,/1X,A,/1X,2(A,1P,E8.1))
99996 FORMAT (1X,2A,1P,E8.1)
99995 FORMAT (1X,A,I5,/1X,A,I5)
      END

      LOGICAL FUNCTION SELCTG(A,B)
*     .. Scalar Arguments ..
*
*     Logical function SELCTG for use with ZGGESX (F08XPF)
*
*     Returns the value .TRUE. if the absolute value of the eigenvalue
*     A/B < 6.0
*
      COMPLEX *16             A, B
*     .. Local Scalars ..
      LOGICAL                 D
*     .. Intrinsic Functions ..
      INTRINSIC               ABS
*     .. Executable Statements ..
      IF (ABS(A).LT.6.0D0*ABS(B)) THEN
         D = .TRUE.
      ELSE
         D = .FALSE.
      END IF
*
      SELCTG = D
*
```

```
      RETURN
      END
```

9.2 Program Data

```
F08XPF Example Program Data
  4                                                                 : Value of N
  (-21.10,-22.50) ( 53.50,-50.50) (-34.50,127.50) (  7.50,  0.50)
  ( -0.46, -7.78) ( -3.50,-37.50) (-15.50, 58.50) (-10.50, -1.50)
  (  4.30, -5.50) ( 39.70,-17.10) (-68.50, 12.50) ( -7.50, -3.50)
  (  5.50,  4.40) ( 14.40, 43.30) (-32.50,-46.00) (-19.00,-32.50) : End of A
  (  1.00, -5.00) (  1.60,  1.20) ( -3.00,  0.00) (  0.00, -1.00)
  (  0.80, -0.60) (  3.00, -5.00) ( -4.00,  3.00) ( -2.40, -3.20)
  (  1.00,  0.00) (  2.40,  1.80) ( -4.00, -5.00) (  0.00, -3.00)
  (  0.00,  1.00) ( -1.80,  2.40) (  0.00, -4.00) (  4.00, -5.00) : End of B
```

9.3 Program Results

```
 F08XPF Example Program Results

 Number of eigenvalues for which SELCTG is true =    2
 (dimension of deflating subspaces)

 Generalized Schur matrix S
                    1                  2                  3                  4
 1  (  10.70, -26.74) ( -72.69, -15.71) (-122.35, -14.08) (  99.00, -38.74)
 2  (   0.00,   0.00) (  11.01,  -3.67) (   4.22,  31.57) ( -19.03, -38.56)
 3  (   0.00,   0.00) (   0.00,   0.00) (  21.04, -63.13) (  12.56,  32.20)
 4  (   0.00,   0.00) (   0.00,   0.00) (   0.00,   0.00) (  21.87, -27.34)

 Generalized Schur matrix T
                    1                  2                  3                  4
 1  (   5.35,   0.00) (  -0.13,  -1.01) (  -1.19,  -3.26) (   4.42,   1.91)
 2  (   0.00,   0.00) (   3.67,   0.00) (  -1.94,   2.21) (   2.90,  -6.17)
 3  (   0.00,   0.00) (   0.00,   0.00) (   7.01,   0.00) (  -2.67,   4.84)
 4  (   0.00,   0.00) (   0.00,   0.00) (   0.00,   0.00) (   5.47,   0.00)

 Matrix of left generalized Schur vectors
                    1                  2                  3                  4
 1  (-0.3733, 0.8687) ( 0.2117,-0.1177) (-0.2156, 0.0104) ( 0.0144,-0.0212)
 2  (-0.1606, 0.0762) (-0.7130,-0.5203) ( 0.1568,-0.3985) (-0.0087,-0.0767)
 3  (-0.1864, 0.0164) (-0.2349, 0.0826) ( 0.2003, 0.6054) (-0.1464,-0.6892)
 4  (-0.0137,-0.1978) ( 0.0473,-0.3131) (-0.5982, 0.0746) (-0.7049,-0.0133)

 Matrix of right generalized Schur vectors
                    1                  2                  3                  4
 1  (-0.9697,-0.2276) ( 0.0340, 0.0612) ( 0.0530,-0.0126) ( 0.0000, 0.0000)
 2  (-0.0052, 0.0023) ( 0.0189,-0.6299) ( 0.7066, 0.3218) ( 0.0000, 0.0000)
 3  (-0.0610,-0.0143) (-0.2882,-0.4647) (-0.4385, 0.0694) ( 0.7034,-0.0728)
 4  ( 0.0143,-0.0610) ( 0.4647,-0.2882) (-0.0694,-0.4385) (-0.0728,-0.7034)

 Reciprocals of left and right projection norms onto
 the deflating subspaces for the selected eigenvalues
 RCONDE(1) =  1.2E-01, RCONDE(2) =  1.6E-01

 Reciprocal condition numbers for the left and right
 deflating subspaces
 RCONDV(1) =  4.8E-01, RCONDV(2) =  4.7E-01

 Approximate asymptotic error bound for selected eigenvalues    =  1.9E-13
 Approximate asymptotic error bound for the deflating subspaces =  4.9E-14
```